First examples on computing resultants in Maple

```
> f := x^2-2;
  g := x^2+1;
```

$$f := x^2 - 2$$

$$g := x^2 + 1$$

```
> with(LinearAlgebra):
> S := SylvesterMatrix(f,g,x);
```

$$S := \begin{bmatrix} 1 & 0 & -2 & 0 \\ 0 & 1 & 0 & -2 \\ 1 & 0 & 1 & 0 \\ 0 & 1 & 0 & 1 \end{bmatrix}$$

```
> Determinant(S);
```

$$9$$

```
> resultant(f,g,x);
```

$$9$$

A second example in $k[x]$ where $k = \mathbb{Q}(y)$

```
> f := x*y-1;
  g := x^2+y^2-4;
```

$$f := x\,y - 1$$

$$g := x^2 + y^2 - 4$$

```
> S := SylvesterMatrix(f,g,x);
```

$$S := \begin{bmatrix} y & -1 & 0 \\ 0 & y & -1 \\ 1 & 0 & y^2 - 4 \end{bmatrix}$$

```
> Determinant(S);
```

$$y^4 - 4\,y^2 + 1$$

```
> R := resultant(f,g,x);
```

$$R := y^4 - 4\,y^2 + 1$$

To apply Proposition 9 of 3.5 to find $A, B \notin \mathbb{Z}[y][x]$ such that $A \cdot f + B \cdot g = R$ where $R = resultant(f, g, x)$ we first solve $s \cdot f + t \cdot g = 1$ for $s, t \notin \mathbb{Q}(y)[x]$ (using the extended Euclidean algorithm) then multiply through by R to clear fractions.

```
> gcdex(f,g,x,'s','t');
```

$$1$$

```
> s;
```

$$-\frac{y\,x}{y^4 - 4\,y^2 + 1} - \frac{1}{y^4 - 4\,y^2 + 1}$$

```
> t;
```

$$\frac{y^2}{y^4-4y^2+1}$$

```
> A := simplify(R*s);
  B := simplify(R*t);
```

$$A := -x\,y-1$$

$$B := y^2$$

```
> A*f+B*g=R;
```

$$(-x\,y-1)\,(x\,y-1)+y^2\,(x^2+y^2-4)=y^4-4\,y^2+1$$

```
> simplify(A*f+B*g-R);
```

$$0$$

We had this rational parametrization of the circle: $x(t)=\dfrac{1-t^2}{1+t^2}$, $y(t)=\dfrac{2\,t}{1+t^2}$. To find the implicit equation f(x,y) = 0 we eliminate t from the polynomials $(1+t^2)\,x-(1-t^2)$ and $(1+t^2)\,y-2\,t$.

```
> t := 't'; # clear assignment to t
```

$$t := t$$

```
> f := (1+t^2)*x-(1-t^2);
  g := (1+t^2)*y-2*t;
```

$$f := (t^2+1)\,x+t^2-1$$

$$g := (t^2+1)\,y-2\,t$$

```
> S := SylvesterMatrix(f,g,t);
```

$$S := \begin{bmatrix} x+1 & 0 & x-1 & 0 \\ 0 & x+1 & 0 & x-1 \\ y & -2 & y & 0 \\ 0 & y & -2 & y \end{bmatrix}$$

```
> Determinant(S);
```

$$4\,x^2+4\,y^2-4$$

```
> resultant(f,g,t);
```

$$4\,x^2+4\,y^2-4$$

Maple uses several algorithms to compute the resultant. It does not expand the determinant of Sylvesters matrix. It turns out that one can modify the Euclidean algorithm to compute the resultant and this is the basis for an efficient algorithm. The basis for this algorithm is covered in exercises 14,15,16,&17 in section 3.5. Maple uses a modification to this algorithm which avoids computing with fractions. I will present the details of the algorithm in the **MACM 401** Introduction to Computer Algebra course in **Spring 2015**.